Быстрое перепроведение и минимизация объемов выгрузки в распределенные базы средствами 1с (для платформы 7.7)

Что происходит при проведении документа 1с? При попадании в процедуру «Обработка проведения», если на документе установлена галочка «Автоматическое удаление движений», то первым делом 1с найдет все текущие движения документа и удалит их. Это потребует «подъема» таблиц движения всех регистров базы (программа не знает заранее, какие именно регистры «двинул» документ), а также таблиц привязанных к движениям документам справочников и периодических реквизитов. Во всех таблицах будет осуществляться поиск и удаление движений и объектов, относящихся к данному документу. Конечно, поиск будет существенно ускорен за счет использования индексных таблиц, и вдобавок, так как все проведение документа выполняется в одной транзакции, то и удаление движений будет проводиться лишь в конце проведения, вместе с операцией записи движений.
Далее, выполняется авторский код программиста по осуществлению нужных движений. Сами движения сразу также не записываются, а лишь формируется общая транзакция. В итоге у движка имеется следующая информация: какие записи из каких таблиц удалить + какие записи в какие таблицы добавить.
Кроме того, если речь идет о распределенной базе, то осуществляется регистрация изменений для документа и всех справочников, которые были удалены/созданы в процессе его проведения.

Затем, если проведение выполнено успешно, транзакция фиксируется – выполняется непосредственная запись/удаление необходимых записей в нужных таблицах.

Все вместе это может отнять существенное время у системы, при этом основную часть временных затрат заберут на себя операции записи и удаления, т.к. их скорость определяется скоростью жесткого диска, а время отклика для жесткого диска может измеряться десятками миллисекунд, что на порядки больше, чем время отклика от оперативной памяти, например. А так как запись идет в различные файлы (таких файлов могут быть десятки, в зависимости от количества задействованных таблиц), то и итоговое количество времени получается существенным (необходимо последовательно найти и изменить разные ячейки памяти на жестком диске, на поиск нужных ячеек и позиционирование головки уйдет много времени). Если время проведения одного документа исчисляется в десятых долях секунды – это уже заметно, а уж если секундами, то время ожидания перепроведения мало-мальски значительного количества документов становится очень длительным.
Кроме того, существенный дискомфорт массовое перепроведение приносит пользователям распределенных баз данных, так как все перепроведенные документы помечаются как измененные, плюс все связанные с их движениями справочники и при последующих операциях выгрузки/загрузки могут довольно долго (в зависимости от размера выгрузки и конфигурации оборудования) записываться в нужные таблицы на базах, принимающих данные изменения.
Массовое перепроведение документов, как правило, необходимо для восстановления последовательности документов. Она нарушается в том случае, если внесены какие либо изменения в уже проведенный документ. В случае перепроведения уже проведенного документа, если граница последовательности находится после позиции данного документа, то граница устанавливается на него и для ее восстановления необходимо будет перепровести все более поздние по позиции документы, относящиеся к данной последовательности.
Анализируя поведение платформы, приходим к выводу, что ряд операций удаления/записи/регистрации изменений (плюс дальнейшей выгрузки/загрузки) будет избыточным, т.к. изменения в одном или нескольких документах последовательности, как правило, могут изменить лишь небольшой процент от всех перепроводимых в процессе ее восстановления документов. А то и вовсе ничего не изменить. Спрашивается, если в движениях документа ничего не изменилось, то нужно ли нам производить все эти действия по удалению/записи/регистрации изменений, создавая тем самым для себя целый ряд неудобств? По-видимому, нет. В данной статье мы рассмотрим приемы, позволяющие избежать избыточности в операциях при восстановлении последовательности документов в 1С версии 7.7.
Для всех документов следует убрать галочку: «Автоматическое удаление движений», но не забыть обязательно использовать функцию: «ОчиститьДвижения()» в модуле проведения в том случае, если будет необходимо перезаписывать движения документа.

Далее, следует завести в глобальном модуле таблицы для каждого из регистров, используемых в конфигурации. Если нет желания использовать глобальные переменные, то можно создавать таблицу с нужной структурой прямо в модуле проведения – это не отнимет много времени. В данные таблицы документы будут записывать свои движения, чтобы затем можно было их сверить с уже существующими движениями документа и определить, следует ли перезаписать движения. Также в случае, когда необходимо новые движения записать, эти таблицы будут служить поставщиками всех необходимых данных для движений. Структура колонок этих таблиц должна в точности повторять структуру измерений, ресурсов и реквизитов регистра. Кроме того, необходима колонка «НомерСтрокиДок» и «Приход», для определения типа движения (Приход/Расход) и номера строки документа, который привязан к данному движению. А также надо создать дополнительные колонки, если в процессе проведения документа из документа нужны какие-либо данные, не являющиеся измерениями, ресурсами или реквизитами регистра. Например, документ может записывать новые значения в периодический справочник «Цены» и т.п.
Допустим в конфигурации у нас описан регистр «Остатки» со следующей структурой:

Измерения:

· Фирма

· Номенклатура

Ресурсы:

· Количество

· ЦенаПост

в глобальном модуле при начале работы системы мы создали таблицу вида:

глТаблДокОстатки = СоздатьОбъект(«ТаблицаЗначений»);

глТаблДокОстатки.НоваяКолонка(«НомерСтрокиДок»);

глТаблДокОстатки.НоваяКолонка(«Приход»);

глТаблДокОстатки.НоваяКолонка(«Фирма»);

глТаблДокОстатки.НоваяКолонка(«Номенклатура»);

глТаблДокОстатки.НоваяКолонка(«Количество»);

глТаблДокОстатки.НоваяКолонка(«ЦенаПост»);
//добавим колонку для периодического реквизита, не участвующего в движениях регистра

глТаблДокОстатки.НоваяКолонка(«ЦенаРеал»);
Далее, в модуле проведения документа «Поступление» (в табличной части у него колонки «Номенклатура», «Количество», «ЦенаПост», «ЦенаРеал», при проведении создает запись в регистре для каждой строки, а также устанавливает в периодических справочниках закупочную и розничную цену), у нас для каждой строки выполняется следующий код:
…

Регистр.Остатки.ПривязыватьСтроку(НомерСтроки);

Регистр.Остатки.Фирма = Фирма;

Регистр.Остатки.Номенклатура = Номенклатура;

Регистр.Остатки.Количество = Количество;

Регистр.Остатки.ЦенаПост = Окр(ЦенаПост*Количество, 2);

Регистр.Остатки.ДвижениеПриходВыполнить();
…
//действия по записи в периодические реквизиты опущены
Этот код мы должны заменить на:

(перед выбором строк документа обязательно: глТаблДокОстатки.УдалитьВсе(); )

глТаблДокОстатки.НоваяСтрока();

глТаблДокОстатки.НомерСтрокиДок = НомерСтроки;

глТаблДокОстатки.Приход = 1;

глТаблДокОстатки.Фирма = Фирма;

глТаблДокОстатки.Номенклатура = Номенклатура;

глТаблДокОстатки.Количество = Количество;

глТаблДокОстатки.ЦенаПост = Окр(ЦенаПост*Количество,2);

глТаблДокОстатки.ЦенаРеал = ЦенаРеал;

Так как каждая строка документа может создавать записи в периодических реквизитах справочника «Цены», то при заполнении таблицы движений мы уже можем попробовать определить – изменится ли текущее значение периодического реквизита, установленное документом, а следовательно, надо ли будет очищать и записывать движения документа. Кроме того, в случае, когда одна и та же номенклатурная позиция встречается в нескольких строках данного документа, нам надо учесть, что значение реквизита справочника после проведения будет установлено по последней строке документа. Для данной проверки используем следующий код:

глЦены.ИспользоватьВладельца(Номенклатура);

//закупочная цена

Если глЦены.НайтиПоРеквизиту("ТипЦен", Константа.ЗакупочнаяЦена, 0) = 0 Тогда


лФлДелатьДвижения = 1;

ИначеЕсли глЦены.Цена <> ЦенаПост Тогда


лФлДелатьДвижения = 1;


//проверим, не создавалось ли уже такое значение документом


лЦенаДок = 0;


глПериодический.ИспользоватьОбъект("Цена", глЦены);


глПериодический.ВыбратьЗначения(ДатаДок, ДатаДок);


Пока глПериодический.ПолучитьЗначение() = 1 Цикл



Если глПериодический.ТекущийДокумент() = ТекущийДокумент() Тогда




лЦенаДок = глПериодический.Значение;




Прервать;



КонецЕсли;


КонецЦикла;


//теперь проверим, совпадают ли значения


Если ЦенаПост = лЦенаДок Тогда



лФлДелатьДвижения = 0;


Иначе



//надо проверить,нет ли в документе еще строк с данной номенклатурой



Если НомерСтроки < КоличествоСтрок() Тогда




//запомним текущие значения




лНомСтар = НомерСтроки;




лНомен = Номенклатура;




//проверим строки до конца




Пока ПолучитьСтроку() = 1 Цикл





Если лНомен = Номенклатура Тогда






Если лЦенаДок = ЦенаПост Тогда
//более поздняя строка установила идентичное //значение – движения не меняем






лФлДелатьДвижения = 0;






Иначе
//более поздняя строка установила другое значение, //движения перезапишем






лФлДелатьДвижения = 1;






КонецЕсли;





КонецЕсли;




КонецЦикла;




//вернем строку на место




ПолучитьСтрокуПоНомеру(лНомСтар);



КонецЕсли;


КонецЕсли;





КонецЕсли;

И аналогичный код для всех других типов цен, которые необходимо установить документом. Если по результатам какой-либо проверки значение лФлДелатьДвижения = 1, то все дальнейшие проверки делать не нужно – движения документа необходимо будет полностью перезаписать.
Допустим, мы полностью сформировали таблицу движений документа, но при этом лФлДелатьДвижения по прежнему равен 0, то есть если изменения и были, то они не коснулись значений цен, установленных документом. Значит, нам надо проверить сами движения как они присутствуют в регистрах. Для этого можно использовать функцию вида: Функция ДелатьДвижДокПоРегистру(Док, ИмяРег), которая для заданного документа будет сверять движения по указанному регистру. Для большей безопасности можно отправлять в параметре саму таблицу полученных движений, но в принципе, достаточно знать лишь имя регистра и автоматически использовать нужную глобальную таблицу. Данная функция должна вернуть 1, если новые движения отличаются от старых, и 0 если они идентичны.
Часть кода самой функции, относящаяся к регистру «Остатки»:

…

ИначеЕсли ИмяРег = «Остатки» Тогда

Рег = СоздатьОбъект("Регистр.Остатки");

//запоминаем количество новых движений

лНомДвиж = глТаблДокОстатки.КоличествоСтрок();

//открываем выборку новых движений

глТаблДокОстатки.ВыбратьСтроки();

//начинаем сверять движения в регистре

Рег.ВыбратьДвиженияДокумента(Док);

Пока Рег.ПолучитьДвижение() = 1 Цикл


Если глТаблДокОстатки.ПолучитьСтроку() = 0 Тогда

//строки в таблице новых движений закончились, а записанные еще нет
Возврат 1;


КонецЕсли;


Если глТаблДокОстатки.НомерСтрокиДок <> Рег.НомерСтроки() Тогда



Возврат 1;


ИначеЕсли глТаблДокОстатки.Приход <> Рег.Приход Тогда



Возврат 1;


ИначеЕсли глТаблДокОстатки.Фирма <> Рег.Фирма Тогда



Возврат 1;


ИначеЕсли глТаблДокОстатки.Номенклатура <> Рег.Номенклатура Тогда



Возврат 1;


ИначеЕсли глТаблДокОстатки.Количество <> Рег.Количество Тогда



Возврат 1;


ИначеЕсли глТаблДокОстатки.ЦенаПост <> Рег.ЦенаПост Тогда



Возврат 1;


КонецЕсли;


лНомДвиж = лНомДвиж - 1;

КонецЦикла;

//проверим общее количество движений – если движения сверились в 0, значит они //абсолютно идентичны
Возврат ?(лНомДвиж = 0,0,1);

ИначеЕсли ИмяРег = …

Таким образом, после заполнения таблицы регистра, в модуле проведения можно использовать следующую проверку:

Если лФлДелатьДвижения = 0 Тогда

лФлДелатьДвижения= глклФункции.ДелатьДвижДокПоРегистру(ТекущийДокумент(), "Остатки");

КонецЕсли;

И далее:

Если лФлДелатьДвижения = 1 Тогда


Если глФлПерепроведение = 1 Тогда



Сообщить(СокрЛП(ТекущийДокумент()) + " - делаем движения...");


КонецЕсли;

//обязательно!

ОчиститьДвижения();


глТаблДокОстатки.ВыбратьСтроки();


Пока глТаблДокОстатки.ПолучитьСтроку() = 1 Цикл



Регистр.Остатки.Фирма = глТаблДокОстатки.Фирма;



Регистр.Остатки.Номенклатура = глТаблДокОстатки.Номенклатура;



Регистр.Остатки.Количество = глТаблДокОстатки.Количество;



Регистр.Остатки.ЦенаПост = глТаблДокОстатки.ЦенаПост;



Регистр.Остатки.ПривязыватьСтроку(глТаблДокОстатки.НомерСтрокиДок);



Если глТаблДокОстатки.Приход = 1 Тогда




Регистр.Остатки.ДвижениеПриходВыполнить();




Иначе




Регистр.Остатки.ДвижениеРасходВыполнить();



КонецЕсли;


//установка периодических реквизитов


…
Теперь в любом случае при проведении документа он будет проводиться только тогда, когда изменились данные о его движениях. Прирост скорости будет заметен, вот пример (при описанной выше структуре регистра/документа): в документе «Поступление» было 2788 строк и обычным способом перепроводился он 3,5 секунды. При новом способе, в случае, когда движения не изменились, перепроведение стало занимать всего 0,2 секунды. Таким образом, прирост скорости перепроведения на данном документе получился в 17,5 раз.

Однако, сохраняется проблема загрузки/выгрузки для распределенных баз. Если использовать стандартную обработку перепроведения, то даже в том случае, если движения документа не изменились, он будет помечен на изменение и попадет в выгрузку. Убирать флаг «Автоматическая регистрация изменений» на объектах конфигурации очень не рекомендуется, если только конфигурация заранее не проектировалась с тем чтобы полностью управлять движением объектов в распределенной системе. А раз так, то нам понадобится собственная обработка перепроведения, которая сама определит, какие документы следует помечать для выгрузки, а какие нет. Благо создать ее не представляет особой сложности. Вот часть кода подобной обработки, отвечающая за регистрацию изменений в проводимом документе:

//установим флаг, по которому модули проведения будут сообщать, делаются ли движения

//проводимым документом

глФлПерепроведение = 1;
лНом = 0; //количество измененных документов для фиксирования транзакции

…

НачатьТранзакцию();

глДок.ВыбратьДокументы();

….

//условия по нужным документам, включая отслеживание необходимости использования
//функции УстановитьТАна(глДок);

…
//отключаем регистрацию изменений для текущего документа

//после перепозиционирования документа регистрация изменений снова будет
//производиться автоматически

глДок.РегистрацияИзменений(0);

Если глДок.Провести(3) = 0 Тогда


ОтменитьТранзакцию();


//надо допровести успешно проведенные документы, код опущен…

…


глФлПерепроведение = 0;


Предупреждение("Не удалось завершить операцию!" + " Расчетный документ: " + СокрЛП(глДок) + ", " + СокрЛП(глДок.ДатаДок) + " (" + СокрЛП(глДок.ПолучитьВремя()) + ")");


Возврат;

Иначе

//флаг глФлПоследнийИзменен устанавливается каждым документом при проведении //и показывает, изменились ли его движения


Если глФлПоследнийИзменен = 1 Тогда



…



//чтобы документ попал в выгрузку, записываем его, включив регистрацию

//изменений
глДок.РегистрацияИзменений(1);


глДок.Записать();



//проверим, надо ли фиксировать транзакцию



лНом = лНом + 1;



Если лНом/100 = Окр(лНом/100) Тогда




ЗафиксироватьТранзакцию();




Сообщить("Зафиксировали транзакцию на отметке: " + СокрЛП(лНом)
 


+ "!");




НачатьТранзакцию();



КонецЕсли;


КонецЕсли;

КонецЕсли;

Как следует из вышеприведенного кода, мы получаем еще один бонус при перепроведении – фактически, все будет проводиться в одной-двух транзакциях, т.к. предполагается, что измененных документов будет немного по отношению к общему числу обработанных документов. Конечно, следует предусмотреть операцию допроведения успешно проведенных документов, в случае ошибки транзакции, но данный код в приведенном примере мы опускаем.
Подводные камни данного метода в том, что в случае изменения структуры регистров, либо логики проведения документа, потребуется вносить соответствующие изменения в код, отвечающий за сверку движений, однако все эти изменения в общем-то тривиальны и требуют лишь аккуратности и внимания, как и любой другой код.
Denis

